# Maintainer: Alexander F. Rødseth <xyproto@archlinux.org>

pkgname=zutty
pkgver=0.14
pkgrel=1
pkgdesc='Terminal emulator'
arch=(loong64 x86_64)
url='https://tomscii.sig7.se/zutty'
license=(GPL3)
depends=(freetype2 libglvnd libxmu xorg-fonts-misc)
makedepends=(git python)
source=('git+https://github.com/tomszilagyi/zutty#commit=bc6e0893e91bc0ffd08cb7d15727973de450c3d6') # tag: 0.14
b2sums=('SKIP')

build() {
  cd $pkgname
  python waf configure --prefix=/usr
  python waf -v build
}

package() {
  cd $pkgname
  python waf --destdir="$pkgdir" install
  install -d "$pkgdir/usr/share/doc/$pkgname"
  cp -v doc/* "$pkgdir/usr/share/doc/$pkgname"
}
